

```

/* 4장 자료의 시각적 표현 */

/* 4.1 chart 프로시저와 gchart 프로시저 */
/* 막대차트, 블록차트, 파이차트, 스타차트 등 작성가능 */

/* goptions 사항들 */
/* goptions    device=win(출력형태를 지정) hsize=? vsize=? (그래프 가로, 세로 크기) horigin=? vorigin=? (가장자리부터의 여백크기)
   border (테두리 선을 지정) ftext=swiss (그리프 전반 글꼴 htext=? (그래프 글자 크기) ftitle=centx (제목을 중앙 위치)
   htitle=? (제목 크기) */

/* axis1 axis2(수평축과 수직축을 지정한다.) */
/* axis1 width=? (막대의 넓이) color=? (색이름) style=n (그래프의 선의 형태 (점선, 실선 등)) label='수평축레이블' */
/* axis2 width=? (막대의 넓이) color=? (색이름) style=n (그래프의 선의 형태 (점선, 실선 등)) label='수직축레이블' */

/*-----*/

data Ex4_1;
input size $ @@;
cards;
medium large large large large medium medium small small
medium medium large small medium large large large small
medium medium medium medium medium large small small
;
run;
/* 빈도 분석 */
proc freq data=Ex4_1;
  table size;
run;
/* 수평막대그래프 */
proc chart data=Ex4_1;
  hbar size;
run;

/*-----*/

data Ex4_2;
input weight @@;
cards;
102.0 106.3 106.6 108.8 107.7 106.1 105.9 106.7 106.8 110.2 101.7
106.6 106.3 110.2 109.9 102.0 105.8 109.1 106.7 107.3 102.0 106.8
110.0 107.9 109.3
;
run;
/* 막대가 4개인 수평막대그래프 */
/* levels=막대의 개수 */
proc chart data=Ex4_2;
  hbar weight / levels=4; /* levels=막대의 개수 */
run;
/* 막대가 8개인 수직막대그래프 */

```

```

proc chart data=Ex4_2;
    vbar weight / levels=8 ;
run;

/* 계급이 4개인 원형그래프 */
proc gchart data=Ex4_2;
    pie weight / levels=4 ;
run;

/*-----*/

data Ex4_3;
input Bakery $ Flavor $ Year Pies_sold @@;
cards;
samford apple 2005 234 samford apple 2006 288
samford blueberry 2005 103 samford blueberry 2006 143
samford cherry 2005 173 samford cherry 2006 195
samford rhubarb 2005 26 samford rhubarb 2006 28
oak apple 2005 219 oak apple 2006 371
oak blueberry 2005 174 oak blueberry 2006 206
oak cherry 2005 226 oak cherry 2006 311
oak rhubarb 2005 51 oak rhubarb 2006 56
clyde apple 2005 213 clyde apple 2006 415
clyde blueberry 2005 177 clyde blueberry 2006 201
clyde cherry 2005 230 clyde cherry 2006 328
clyde rhubarb 2005 60 clyde rhubarb 2006 59
;
run;

/* Flavor, Pies_sold의 수직막대그래프 */
/* subgroup=각각의 기호를 표시, sumvar=y축의 변수*/
proc chart data=Ex4_3;
    vbar Flavor / subgroup=Bakery sumvar=Pies_sold ;
run;

proc gchart data=Ex4_3;
    vbar Flavor / subgroup=Bakery sumvar=Pies_sold ;
run;

/*-----*/

data Ex4_4;
input gender $ age chol @@;
cards;
M 23 40 M 64 88 M 66 110 M 31 86 M 55 137 M 48 78 M 31 46
M 58 11 M 31 88 M 27 80 M 25 66 M 23 65 F 40 60 F 38 53
F 56 89 F 34 90 F 78 110 F 45 100 F 20 89 F 56 78 F 44 99
F 30 89 F 24 57 F 89 100 F 80 90 F 22 60
;
run;

/* 성별에 따른 수직막대그래프 */
/* width=막대의 너비 space=막대 사이의 너비*/

```

```

proc chart data=Ex4_4;
    vbar gender / width=10 space=10;
run;
/* 연령에 따른 8개의 수평막대그래프*/
/* levels=막대의 개수*/
proc chart data=Ex4_4;
    hbar age / levels=8;
run;
/* 연령에 따른 도넛 그래프*/
/* donut 함수는 gchart에서만 가능하다.*/
proc gchart data=Ex4_4;
    donut age / levels=8;
run;

/*-----*/

/* 4.2 plot 프로시저와 gplot 프로시저 */

/*-----*/

/* 변수eng의 길이를 5개로 지정 */
data Ex4_5;
length eng $5;
input eng dollars num;
cards;
Civil 27308 73273
Areo 29844 70192
Elec 22920 89382
Mech 32810 19620
Chem 28116 25541
Petro 18444 34923
;
run;
/* axis 축지정 */
/* offset=(시작위치, 끝위치) */
axis1 offset=(10,10);
/* 버블 산점도 그래프*/
/* gplot에서만 가능, haxis=수평축지정 */
proc gplot data=Ex4_5;
    format dollars dollar9. ;
    bubble dollars*eng=num / haxis=axis1;
run;

/*-----*/

data Ex4_6;
input weight fuel @@;
cards;
3.4 5.5 3.8 5.9 4.1 6.5 2.2 3.3 2.6 3.6

```

```

2.9 4.6 2.0 2.9 2.7 3.6 1.9 3.1 3.4 4.9
;
run;
/* 산점도 구하기 */
/* plot x축변수*y축변수='모양' */
proc gplot data=Ex4_6;
    plot fuel*weight='star';
run;

/*-----*/

data Ex4_7;
input weight num @@;
cards;
8.2 2 8.3 6 8.4 8 8.5 11 8.6 17
8.7 17 8.8 24 8.9 18 9.0 13 9.1 10 9.2 4
;
run;
/* goptions 작성 */
/* goptions 사항들 */
/* goptions device=win (출력형태를 지정) hsize=? vsize=? (그래프 가로 크기) horigin=? vorigin=? (가장자리부터의 여백크기)
border (테두리 선을 지정) ftext=swiss (그리프 전반 글꼴) htext=? (그래프 글자 크기) ftitle=centx (제목을 중앙 위치) htitle=? (제목
크기)
axis1 axis2(수평축과 수직축을 지정한다.) */
goptions device=win hsize=5 vsize=5 horigin=1 vorigin=1 ftext=swiss htext=1.0 ftitle=centx
htitle=1;
/* symbol 작성*/
/*symbol h=(크기) v=(기호) i=join (연결해줌); */
symbol1 h=0.5 v=dot i=join;
/* 도수꺾은선 그래프 작성*/
proc gplot data=Ex4_7;
    plot num*weight=1;
run;

/*-----*/

data Ex4_8;
input weight nums @@;
cards;
8.2 2 8.3 6 8.4 8 8.5 11 8.6 17
8.7 17 8.8 24 8.9 18 9.0 13 9.1 10 9.2 4
;
run;
/* goptions과 symbol의 작성 */
goptions device=win hsize=5 vsize=5 horigin=1 vorigin=1 ftext=swiss htext=1.0 ftitle=centx
htitle=1;
symbol1 h=0.5 v=dot i=join;
/* 추정회기 직선 작성*/
/*

```

```

proc reg data=Ex4_8;
    model nums=weight ;
    output out=b p=yhat;
run;

/* 산점도랑 추정회귀직선 겹쳐서 작성*/
/* overlay 겹치게해줌 */

proc gplot data=Ex4_8;
    plot nums*weight='star' yhat*weight=1 / overlay ;
run;

/*-----*/

/* 4.3 boxplot 프로시저 */

/*-----*/

data Ex4_9;
input id name $ gender $ dept $ rep cha min fin;
total = rep+cha+min+fin;
cards;
023401 SK.KIM M STAT 10 20 40 34
023405 SI.HWANG M STAT 10 18 38 28
023410 KS.LEE F MATH 10 19 38 40
023411 SS.LEE M MATH 10 20 29 34
023412 MS.SEO F ECON 10 20 38 30
023415 NR.JUNG F ECON 10 15 20 34
023420 KW.YEON M STAT 10 19 34 35
023425 HJ.KIM F STAT 10 20 20 28
023426 NK.LEE M ECON 10 20 5 10
023428 JS.BAE M MATH 10 20 38 40
;
run;

/*boxplot는 먼저 정렬이 필요하다.*/

proc sort data=Ex4_9;
    by gender;
run;

/*boxplot 작성*/
/* boxstyle=(schematic 또는 skeletal) boxwidth=박스폭 hoffset=상자간의 간격 */

proc boxplot data=Ex4_9;
    plot total*gender / boxstyle=schematic boxwidth=20 hoffset=15;
run;

/*-----*/

```